TwoNonIsomorphicExtensions

Let $R := \mathbb{Q}[x, y, z]$. In this example we want to study two non-equivalent extensions $0 \to K \to M \to L \to 0$ and $0 \to K \to N \to L \to 0$ with K a torsion module and L a torsion free module. Our goal is to use the notion of functor to reveal that these extensions are not only non-equivalent but also non-isomorphic. For this we define two functors F_M and F_N and study their behavior when applied to complexes. Concretely, we apply F_M resp. F_N to $0 \to K \to M \to L \to 0$ which we refer to as

$$S : 0 \to K \xrightarrow{\alpha_1} M \xrightarrow{\alpha_2} L \to 0.$$

In the following we consider

$$F_Q = \mathtt{Ext}(1, -, Q) = \mathrm{Ext}^1(-, Q) = R^1 \operatorname{Hom}_R(-, Q)$$

```
> restart;
> with(Involutive): with(homalg):
```

Specify the `homalg`-table of the ring package `Involutive`:

```
> RPI:=‘Involutive/homalg‘;
```

$$RPI := Involutive/homalg$$

Use the ring package `Involutive` as the default ring package:

```
> ‘homalg/default‘:=RPI;
```

$$homalg/default := Involutive/homalg$$

We force the Maple ring package `Involutive` to use the external program `JB`, which is a C++ implementation of the involutive division algorithm:

```
> InvolutiveOptions("C++");
```

Define the ring $R = \mathbb{Q}[x, y, z]$:

```
> var:=[x,y,z];
```

$$var := [x, y, z]$$

The two presentation matrices `M` and `N`:

```
> MM:=matrix([[x*z, z*y, z^2, 0, 0, y], [0, 0, 0, z^2*y-z^2, z^3, x*z],
[0, 0, 0, z*y^2-z*y, z^2*y, x*y], [0, 0, 0, x*z*y-x*z, x*z^2, x^2],
[-x*y, -y^2, -z*y, x^2*y-x^2-y+1, x^2*z-z, 0], [x^2*y-x^2, x*y^2-x*y,
x*z*y-x*z, -y^3+2*y^2-y, -z*y^2+z*y, 0]]);
```

$$MM := \begin{bmatrix} x\,z & z\,y & z^2 & 0 & 0 & y \\ 0 & 0 & 0 & z^2\,y - z^2 & z^3 & x\,z \\ 0 & 0 & 0 & z\,y^2 - z\,y & z^2\,y & x\,y \\ 0 & 0 & 0 & x\,z\,y - x\,z & x\,z^2 & x^2 \\ -x\,y & -y^2 & -z\,y & x^2\,y - x^2 - y + 1 & x^2\,z - z & 0 \\ x^2\,y - x^2 & x\,y^2 - x\,y & x\,z\,y - x\,z & -y^3 + 2\,y^2 - y & -z\,y^2 + z\,y & 0 \end{bmatrix}$$

```
> NN:=matrix([[x*y, y^2, z*y, 0, 0, z], [0, 0, 0, z*y^2-z*y, z^2*y,
x*z], [x^2*z, x*z*y, x*z^2, -z^2*y+z^2, -z^3, 0], [0, 0, 0,
y^3-2*y^2+y, z*y^2-z*y, x*y-x], [0, 0, 0, x^2*y-x^2-y+1, x^2*z-z, y],
[x^3, x^2*y, x^2*z, -x*z*y+x*z, -x*z^2, 0]]);
```

$$NN := \begin{bmatrix} x\,y & y^2 & z\,y & 0 & 0 & z \\ 0 & 0 & 0 & z\,y^2 - z\,y & z^2\,y & x\,z \\ x^2\,z & x\,z\,y & x\,z^2 & -z^2\,y + z^2 & -z^3 & 0 \\ 0 & 0 & 0 & y^3 - 2\,y^2 + y & z\,y^2 - z\,y & x\,y - x \\ 0 & 0 & 0 & x^2\,y - x^2 - y + 1 & x^2\,z - z & y \\ x^3 & x^2\,y & x^2\,z & -x\,z\,y + x\,z & -x\,z^2 & 0 \end{bmatrix}$$

The two modules $M := \operatorname{coker}\left(R^{1\times 6} \xrightarrow{\mathtt{M}} R^{1\times 6}\right)$ and $N := \operatorname{coker}\left(R^{1\times 6} \xrightarrow{\mathtt{N}} R^{1\times 6}\right)$:

```
> M:=Cokernel(MM,var);
```

$$M := [[[1, 0, 0, 0, 0, 0] = [1, 0, 0, 0, 0, 0], [0, 1, 0, 0, 0, 0] = [0, 1, 0, 0, 0, 0], [0, 0, 1, 0, 0, 0] = [0, 0, 1, 0, 0, 0], [0, 0, 0, 1, 0, 0] = [0, 0, 0, 1, 0, 0], [0, 0, 0, 0, 1, 0] = [0, 0, 0, 0, 1, 0], [0, 0, 0, 0, 0, 1] = [0, 0, 0, 0, 0, 1]], [[x\,z, z\,y, z^2, 0, 0, y], [0, 0, 0, z^2\,y - z^2, z^3, x\,z], [0, 0, 0, z\,y^2 - z\,y, z^2\,y, x\,y], [0, 0, 0, x\,z\,y - x\,z, x\,z^2, x^2], [x^2\,z, x\,z\,y, x\,z^2, 0, 0, x\,y], [-x\,y, -y^2, -z\,y, x^2\,y - x^2 - y + 1, x^2\,z - z, 0], [x^2\,y - x^2, x\,y^2 - x\,y, x\,z\,y - x\,z, -y^3 + 2\,y^2 - y, -z\,y^2 + z\,y, 0], [0, 0, 0, z\,y - z, z^2, x^3 - y^2]], \text{“Presentation”}, \frac{3\,s^2}{1-s} + \frac{2\,s}{1-s} + \frac{2}{(1-s)^2} + \frac{3}{(1-s)^3} + \frac{2\,s^2}{(1-s)^2} + s^2 + s + \frac{1}{1-s} + \frac{s}{(1-s)^2}, [34, 14, 3]]$$

```
> N:=Cokernel(NN,var);
```

$$N := [[[1, 0, 0, 0, 0, 0] = [1, 0, 0, 0, 0, 0], [0, 1, 0, 0, 0, 0] = [0, 1, 0, 0, 0, 0], [0, 0, 1, 0, 0, 0] = [0, 0, 1, 0, 0, 0], [0, 0, 0, 1, 0, 0] = [0, 0, 0, 1, 0, 0], [0, 0, 0, 0, 1, 0] = [0, 0, 0, 0, 1, 0], [0, 0, 0, 0, 0, 1] = [0, 0, 0, 0, 0, 1]], [[x\,y, y^2, z\,y, 0, 0, z], [0, 0, 0, z\,y^2 - z\,y, z^2\,y, x\,z], [x^2\,z, x\,z\,y, x\,z^2, -z^2\,y + z^2, -z^3, 0], [0, 0, 0, y^3 - 2\,y^2 + y, z\,y^2 - z\,y, x\,y - x], [0, 0, 0, x^2\,y - x^2 - y + 1, x^2\,z - z, y], [x^2\,y, x\,y^2, x\,z\,y, 0, 0, x\,z], [x^3, x^2\,y, x^2\,z, -x\,z\,y + x\,z, -x\,z^2, 0], [0, 0, 0, x\,y^2\,z - x\,z\,y, x\,y\,z^2, x^2\,z], [0, 0, 0, 0, 0, x^3\,z - z\,y^2 - x\,z], [0, 0, 0, x\,y^3 - 2\,x\,y^2 + x\,y, x\,y^2\,z - x\,z\,y, x^2\,y - x^2], [0, 0, 0, 0, 0, x^3\,y - x^3 - y^3 - x\,y + y^2 + x]], \text{“Presentation”}, \frac{3\,s}{1-s} + \frac{2}{(1-s)^2} + 2\,s^2 + \frac{3}{(1-s)^3} + \frac{2\,s^2}{(1-s)^2} + \frac{s^2}{1-s} + s^3 + \frac{1}{1-s} + \frac{s^3}{1-s} + \frac{s}{(1-s)^2}, [51, 17, 3]]$$

The torsion submodule $K := t(M)$ of M (and N):

```
> K:=TorsionSubmodule(M,var);
```

$$K := [[[1, 0, 0] = [0, 0, 0, 0, 0, 1], [0, 1, 0] = [0, 0, 0, y-1, z, 0], [0, 0, 1] = [x, y, z, 0, 0, 0]], [[y, 0, z], [0, -z\,y, x\,z], [x\,z, z^2, 0], [0, -y^2 + y, x\,y - x], [x\,y, z\,y, 0], [0, x^2 - 1, -y], [x^2, x\,z, 0]], \text{“Presentation”}, s + \frac{1}{1-s} + \frac{s}{(1-s)^2} + \frac{2}{(1-s)^2} + \frac{s}{1-s}, [7, 3, 0]]$$

The embedding $0 \to K \xrightarrow{\alpha_1} M$:

```
> alpha1:=TorsionSubmoduleEmb(M,var);
```

$$\alpha 1 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & y-1 & z & 0 \\ x & y & z & 0 & 0 & 0 \end{bmatrix}$$

The torsion free factor $L := M/K = M/t(M)$ of M (and N):

```
> L:=Cokernel(alpha1,M,var);
```

$$L := [[[1, 0, 0, 0, 0] = [1, 0, 0, 0, 0, 0], [0, 1, 0, 0, 0] = [0, 1, 0, 0, 0, 0], [0, 0, 1, 0, 0] = [0, 0, 1, 0, 0, 0], [0, 0, 0, 1, 0] = [0, 0, 0, 1, 0, 0], [0, 0, 0, 0, 1] = [0, 0, 0, 0, 1, 0]], [[0, 0, 0, y-1, z], [x, y, z, 0, 0]], \text{“Presentation”}, \frac{2}{(1-s)^2} + \frac{3}{(1-s)^3}, [5, 5, 3]]$$

The natural epimorphism $M \xrightarrow{\alpha_2} L \to 0$:

```
> alpha2:=CokernelEpi(alpha1,M,var);
```

$$\alpha 2 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

The short sequence $S : 0 \to K \xrightarrow{\alpha_1} M \xrightarrow{\alpha_2} L \to 0$ is exact:

```
> IsShortExactSeq(K,alpha1,M,alpha2,L,var);
```

$$true$$

The functor $\mathrm{Ext}^1_R(-, N)$ applied to the short exact sequence S:

```
> coseqN:=ExtOnSeqs(1,N,[alpha2,alpha1],[L,M,K],var):
```

This says that the map $\mathrm{Ext}^1_R(L, N) \xrightarrow{\mathrm{Ext}^1_R(\alpha_2,N)} \mathrm{Ext}^1_R(M, N)$ is not injective, i.e. that the connecting homomorphism $\mathrm{Hom}_R(K, N) \xrightarrow{\delta^0} \mathrm{Ext}^1_R(L, N)$ is non-trivial:

```
> IsShortExactSeq(op(MakeCoseq(coseqN)),var,"VERBOSE");
```

$$\text{“homs”} = true,\ \text{“cmps”} = true,\ \text{“defs”} =$$

$$[[[1 = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}],\ [z,\ y,\ x],\ \text{“Presentation”},\ 1,\ [0,\ 0,\ 0]],\ true,\ true]$$

The functor $\mathrm{Ext}^1_R(-, M)$ applied to the short exact sequence S:

```
> coseqM:=ExtOnSeqs(1,M,[alpha2,alpha1],[L,M,K],var):
```

Whereas the resulting sequence $\mathrm{Ext}^1_R(S, M)$ is again exact. In particular, the connecting homomorphism $\mathrm{Hom}_R(K, M) \xrightarrow{\delta^0} \mathrm{Ext}^1_R(L, M)$ is trivial, i.e. the zero map:

```
> IsShortExactSeq(op(MakeCoseq(coseqM)),var,"VERBOSE");
```

$$true$$

Author: Mohamed Barakat

Date:	2007-05-22
Last modified:	2007-07-27 12:17

References

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* Janet*:* I. *Polynomial Systems.* II. *Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (http://wwwb.math.rwth-aachen.de/Janet).

[BR] Mohamed Barakat and Daniel Robertz, homalg – *A meta-package for homological algebra*, submitted. arXiv:math.AC/0701146 and (http://wwwb.math.rwth-aachen.de/homalg).

[BR07] ———, homalg *project*, 2004-2007, (http://wwwb.math.rwth-aachen.de/homalg).

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany

E-mail address: mohamed.barakat@rwth-aachen.de